
Notes on a GAMS adaptation of the SGMCGE model 

 

During the DOE integrated assessment project a GAMS version of the SGM model has 
been built (SGMGAMS for short).  SGMGAMS is a static single year (1990), single country 
(US) representation which uses the data from the total SGM model.  The data used in 
SGMGAMS are mainly from the **usa90cal22j.xls** obtained from the SGM developers 
(Sands).   

 
SGMGAMS includes 23 sectors.  Within these sectors subsectors are included in the 

electricity generation sector based on the type of fuel source (generation using oil, coal, gas, 
biomass, nuclear, and hydropower).  On the factor side SGMGAMS depicts use and productivity 
of 3 traditional factors (labor, capital, and land) and 22 intermediate inputs.  In addition, there are 
4 vintages of capital stocks represented. 

 
While we have made efforts to completely mimic we have not fully included the full 

SGM features on the carbon market, the biomass subsector, and agricultural lands as we have not 
fully tried to incorporate the AGLU features due to it’s potential differences with our sector 
based estimation approach.  We feel a need to have more interaction with the SGM developers to 
finish this part. 
 
Overview of SGMGAMS in GAMS 
 

The basic components of a GAMS model consist of SETS, PARAMETERS, 
VARIABLES, and EQUATIONS.   
 
Sets 
 

Sets are the basic building a block of GAMS and define the model scope.  SGMGAMS 
contains a number of SETS.  The most important SETS mainly used in the model are 
 

Sector which defines the 23 producing sector with the elements 
{OtherAg, CrudeOil, NatGas, Coal, Electricity, … , FoodProcess} 

SubSector which defines the fuel based subsectors 
{Oil, Gas, Coal, Biomass, Nuclear, Hydro} 

Activity which identifies the type of inputs that can be used 
{OtherAg, Service, CrudeOil, … , Labor, Land, Capital} 

Vintage which identifies the number of periods ago that the item was installed 
 {-3, -2, -1, 0} 

 
In addition, the SET called “AllSet” is used to insure proper ordering and includes the elements 
from all other SETS used in the model.   
 
The complete SETS definition and declaration are shown from lines 63 to 1214 in the allsgm.lst 
file.   

Parameters 
 

Input data can be derived directly from the given data entry (i.e. Tables, Parameters, or 
Scalars as shown from lines 1246 to 1961) or from the direct assignment through a calculation 
within the model (lines 3603 to 4711, and lines 4926 to 7073).   

 
Direct assignments are mostly used in calibrating parameters needed in the model.  The 

following is an example of CES value-added share factor input parameter calculation: 
 
DeltaVa(Factor2,AllSecAllSub) 
  $(  (QF0(Factor2,AllSecAllSub) ne 0) 
    and YesCESVa(AllSecAllSub) 
   ) 
   = ( FactPriceDist0(Factor2,AllSecAllSub) 
         * FactPricNation(Factor2) 
            * (QF0(Factor2,AllSecAllSub))**(1 + RhoVa(AllSecAllSub)) 
      ) /SUM(Factor2P, 
               FactPriceDist0(Factor2P,AllSecAllSub) 
                 * FactPricNation(Factor2P) 
                     *(QF0(Factor2P,AllSecAllSub))**(1 + RhoVa(AllSecAllSub)) 
             ) ; 
 
The parameters used in the calibration process are defined in sgmcalib.gms and are shown from 
line 4792 to 4921 in the allsgm.lst file.   
 
Variables 
 

The decision variables or endogenous variables can be unrestricted in sign (i.e. 
government saving that can be negative depicting government borrowing) or constrained to be 
nonnegative (or in the GAMS language POSITIVE).  These variables are declared in 
sgmmodel.gms and in the lines listed as 7278 to 7344 in the allsgm.lst file and use the ALLSET 
and it’s aliases here we include more information on what goes in the index positions as follows 
 
VARIABLES 

GovSaving    Government saving or borrowing 
 
POSITIVE VARIABLES 
* #### Quantity block  

FactorQ(Factor,Sector,SubSector) Factor use by a producing sector for CES bottom level 
Production(Sector,SubSector,Vintage) Production quantity 
QValAdd(Sector,SubSector) Quantity value-added 
QintA(Sector,SubSector) Quantity intermediate inputs 
QIntC(Activity,Sector,SubSector) Intermediate demand 
QExport(Sector,SubSector)  US export quantity 
QImport(Sector,SubSector)  US import quantity 

* #### Price block 

FactPriceNation(Factor)   Average economy-wide wage average 
FactPriceSec(Factor,Sector,SubSector) Factor price by sector and subsector 
ComPrice(Sector)    Prices of commodities 
PricePaid(Activity,AllSecAllSub) Prices paid to input i for output j 
PriceReceive(AllSecAllSub)   Prices received from output j 
PValAdd(Sector,SubSector)  Value-added price 
PIntA(Sector,SubSector)   Intermediate inputs price 
PExp(Commodity) Price received by producers for exported goods 
PImp(Commodity)   Price paid by consumer for imported goods 
PCarb(Region)    Carbon price 

 
* #### Household consumption information 

GoodsDemand(Commodity,HouseholdH) Commodity demand by households 
HhExpend(HouseholdH)   Household expenditure 
HHIncome(HouseholdH)   Household income 

 
* #### Investment, expected profit rate, and capital stock information 

AlphaInv1(Activity,AllSecAllSub) Dynamic calculation for expected profit rate 
AlphaInv2(AllSecAllSub)  Dynamic calculation for expected profit rate 
AlphaInv3(Activity,AllSecAllSub) Dynamic calculation for expected profit rate 
AlphaInv(Activity,AllSecAllSub)  Dynamic calculation for expected profit rate 
IBigAij(Activity,AllSecAllSub)             Dynamic calculation for expected profit rate 
QInvest(AllSecAllSub)                    Quantity investment 
ExpProfSec(Sector)                  Expected profit rate for investment by Sector 
ExpProfSub(SubSector)                          Expected profit rate for investment by SubSector 
Zvalue(AllSecAllSub)                   Aggregate calculation for Z  
SubInvSh1(AllSecAllSub)                    Aggregate calculation for share investment page 25 
SubInvSh3(AllSecAllSub)  Aggregate calculation for share investment page 25 
KStock(AllSecAllSub,Vintage)                Implied capital 

 
* #### Government information 

GovExpend    Government expenditure 
GovTrnsfPaymt    Government transfer payments 
GovDemand(HouseholdG)  Government demands for goods and services 
GovIncome    Total government tax revenue 
GovConsumption(Sector,HouseHoldG)  Government consumption 

 
* #### Greenhouse gas information 

AgSinkQ(Ghg,Vintage)   Agricultural sinks using response functions 
AgEmitQ(Ghg,Vintage)   Agricultural sources using response functions 
EnergyEmitQ(Sector,Ghg,Vintage) Energy emissions 
NetEmiTQ(Region,Ghg)   Net emissions 
CarbTaxRate(Sector,Ghg)  Carbon tax rate 
CarbTaxTot(Sector,Ghg) Total Carbon tax revenues collecting from sectors 

Equations 
 

Equations represent groups of relationships in the model more than one equation 
depending on how many domains (indices) are in equations.  The equations are defined as 
follows: 
 

* #### Quantity block 
CESQVAEq(Sector,SubSector)  Quantity value-added for the bottom level CES  
CESQVAFOC(Factor)   Factor demand from the first order condition 
FactorMkt(Factor,Sector,SubSector) Factor market balances 
QValAddEq(Sector,SubSector)  Quantity value-added for the top level Leontief 
QintAEq(Sector,SubSector)  Quantity intermediate inputs for the Leontief  
QIntCEq(Activity,Sector,SubSector) Intermediate demand  
ImpBal(Commodity)   Import balance 
ExpBal(Commodity)                      Export balance 
ExportEq(Sector,SubSector)                   US export quantity in billion 
ImportEq(Sector,SubSector)                   US import quantity in billion 
 

* #### Price block 
FactPriceSecEq(Factor,Sector,SubSector) Factor price at sector and subsector levels 
PVaEq(Sector,SubSector) Activity revenue and costs where Revenue is 

exhausted by payments for value-added and 
intermediate inputs 

PIntAEq(Sector,SubSector)                     Aggregate intermediate inputs price 
PricePaidEq(Activity,AllSecAllSub)  Price paid to input i equation 
PriceReceivEq(AllSecAllSub)                  Price received from output j 
 

* #### Household consumption information 
GoodsDemandEq(Commodity,HouseholdH) Commodity demand by Households 
HhExpendEq(HouseholdH)  Household expenditure balance 
IncomeEq(HouseholdH)                         Household budget constraint 
 

* #### Investment, expected profit rate, and capital stock information 
AlphaInv1Eq(Activity,AllSecAllSub) Dynamic calculation for expected profit rate 
AlphaInv2Eq(AllSecAllSub)  Dynamic calculation for expected profit rate 
AlphaInv3Eq(Activity,AllSecAllSub) Dynamic calculation for expected profit rate 
AlphaInvEq(Activity,AllSecAllSub) Dynamic calculation for expected profit rate 
IBigAijEq(Activity,AllSecAllSub) Dynamic calculation for expected profit rate 
InvestDem(AllSecAllSub)  Investment demand balance 
ExpProfSecEq(Sector)   Expected profit rate for investment by Sector 
ExpProfSubEq(SubSector)  Expected profit rate for investment by SubSector 
ZvalueEq(AllSecAllSub)   Aggregate calculation for Z 
SubInvSh1Eq(AllSecAllSub)  Aggregate calculation for share investment page 25 
SubInvSh3Eq(AllSecAllSub)  Aggregate calculation for share investment page 25 

KStockEq(AllSecAllSub,Vintage) Implied capital balance 
 

* #### Equilibrium conditions 
Profit(Sector,SubSector,Vintage)             Zero profit condition 
SupplyDemandEq(Sector)                      Commodity market balance 
 

* #### Government information 
GovDemandEq(HouseHoldG)                 Government demands for goods and services 
GovTrnsfEq                                   Government transfer payment equation 
GovIncomeEq                                 Government net tax revenue equation 
GovBudgetEq                                 Government budget constraint 
GovConsumpEq(Sector,HouseHoldG) Government consumption equation 
GovExpendEq                               Government expenditure 
 

* #### Greenhouse gas emissions and sinks information 
AgEmitQEq(Ghg,Vintage)               Agricultural source emissions 
AgSinkQEq(Ghg,Vintage)             Agricultural sinks 
EnergyEmitqEq(Sector,Ghg,Vintage) Energy emissions 
NetEmisQEq(Region,Ghg)  Net emissions 
EmisQLimitEq(Region)                        Emission limits constraint 
CarbTaxRateEq(Sector,Ghg)            Carbon tax rate 
CarbTaxTotEq(Sector,Ghg)              Total Carbon tax revenues collecting from sectors 
 

Model Structure 
 

In general, the structure of SGMGAMS mirrors the SGM model structure.  Details 
follow: 
 
Production Sector 

 
When intermediate inputs are used in the production of goods, the production process is 

modeled as 2-level nested employing CES and Leontief production functions.  The top level 
including the aggregated intermediate inputs and value-added is represented using a Leontief 
production process whereas the bottom level models the aggregated intermediate inputs as the 
Leontief technology of all intermediate inputs but the value-added is represented by the CES 
production function of the production factors (labor and capital). 
 

X Final Goods 
                                   -------------------------------------------------------- 
              |       | 
              |       |       => Leontief 
                              Intermediate Inputs                 Value-Added 
                        -------------------------------                      --------------------- 
                        |        |           |       | => CES 
Leontief <=     |                     |           |       | 

         Interm. Input1           Interm. Input2  Capital  Labor    

Five equations involved in the production process and input use are: 
 
Quantity Value-Added for the CES Bottom Level Technology 

The value-added is represented by the CES production function of the production factors 
(labor and capital) for the bottom level. 
 
CESQVAEq(Sector,SubSector) 
    $(     TisMap(Sector,SubSector) 
       and YesCESVa(Sector) 
     ).. 
 
   QValAdd(Sector,SubSector) 
    =E= 
   AlphaVa1(Sector,SubSector) 
    *(SUM(Factor2, 
             DeltaVa1(Factor2,Sector,SubSector) 
              * FactorQ(Factor2,Sector,SubSector) **(-RhoVa1(Sector,SubSector)) 
          ) 
      )**(-1/RhoVa1(Sector,SubSector))    ; 
 
Factor Demand 
  

This first order condition implies that the marginal factor cost (labor and capital) is equal 
to the marginal revenue product (net of intermediate input costs) of the factor.  This equation 
follows equation (16) in Lofgren 2001.1 
 
CESQVAFOC(Factor2) 
    $( sum(Sector$YesCESVa(Sector), 
            sum(MapSecSub(Sector,SubSector), 
                 FactPriceDist1(Factor2,Sector,SubSector) )) 
      ).. 
 
   FactPriceNation(Factor2) 
   =E= 
   sum(Sector$YesCESVa(Sector), 
     sum(MapSecSub(Sector,SubSector), 
           (  PValAdd(Sector,SubSector)*(1 - ValAddTax1(Sector,SubSector)) 
               *QValAdd(Sector,SubSector) 
                    * SUM(Factor2P, 
                            DeltaVa1(Factor2P,Sector,SubSector) 
                             * FactorQ(Factor2P,Sector,SubSector)**(-RhoVa1(Sector,SubSector)) 
                           )**(-1) 
                              * DeltaVa1(Factor2,Sector,SubSector) 
                                                 
1 Lofgren, H., R. L. Harris, S. Robinson, M. Thomas, and M. El-Said, “A Standard Computable General Equilibrium 
(CGE) Model in GAMS, TMD Discussion Paper No. 75, IFPRI, May 2001. 

                                 * FactorQ(Factor2,Sector,SubSector)**(-RhoVa1(Sector,SubSector)-1) 
            ) 
        )) 
  /sum(Sector$YesCESVa(Sector), 
     sum(MapSecSub(Sector,SubSector), 
      FactPriceDist1(Factor2,Sector,SubSector) )) 
   ; 
 
Quantity Value-Added for the Leontief Top Level Technology 

 
The value-added is represented by the Leontief technology for the top level. 

 
  QValAdd(Sector,SubSector) 
     =E= 
  iva1(Sector,SubSector) 
    *sum(Vintage,Production(Sector,SubSector,Vintage)) 
       ; 
 
Quantity Intermediate Inputs 
 

When the intermediate inputs are used in the production of goods, it is modeled as a 
Leontief function at the lower level.  IntA1(Sector,SubSector) is the input-output coefficient 
which is calculated as the quantity of input divided by the quantity of output. 
 
QintAEq(Sector,SubSector) 
  $TisMap(Sector,SubSector).. 
 
  QintA(Sector,SubSector) 
     =G= 
  IntA1(Sector,SubSector) 
   * sum(Vintage,Production(Sector,SubSector,Vintage) ) 
       ; 
 
Quantity Intermediate Input Demand 

 
The demand for disaggregated intermediate inputs is modeled as a Leontief production.  

ica1 represents a fixed intermediate input coefficient and is calculated within the model (see lines 
5214 to 5223 and lines 6053 to 6056). 
 
QIntCEq(Activity,Sector,SubSector) 
  $(    TisMap(Sector,SubSector) 
    and ica1(Activity,Sector,SubSector) 
   ).. 
 
  QIntC(Activity,Sector,SubSector) 
    =G= 

  ica1(Activity,Sector,SubSector) 
     * QIntA(Sector,SubSector)     ; 
 
Household Sector 

 
Factors such as population, labor and capital endowment, personal saving, and transfer 

payments are involved in modeling the household consumption.  The population data is given by 
SGM data. 
 
Household Income 

 
The household income is the sum of retained earning income, labor income, government 

transfer payments minus personal income tax and saving.  These household income and saving 
equations follow equations (33) – (40) on pages 28-30 in the SGM model documentation.  
Because of space limitation, these equations are not presented here but the completed expression 
of these equations involved the household consumption can be looked up from lines 7771 to 
7857. 

 
Household Budget 

 
The household budget is used to make sure that the incomes after tax minus personal 

saving plus the government transfer payments are exhausted (see lines 7714 to 7769). 
 

Household Consumption 
 
The household consumption is allocated across different commodities according to the 

Linear Expenditure System (LES): 









−+= ∑

i
iiiiiii PYPQP γβγ  

where  is i commodity price, Q  is i commodity quantity,  is the expenditure on i 
commodity,  is the marginal budget share,  is the minimum requirements subsistence 
quantity, and Y is the income.  Below shows how this LES is incorporated in the model 

iP i iiQP

iβ iγ

 
GoodsDemandEq(Commodity,HouseholdH) 
  $(     Beta(Commodity,HouseHoldH) 
     and ComPrice0(Commodity) 
     and SAM(Commodity,HouseholdH) 
   ).. 
 
    ComPrice(Commodity) 
        *GoodsDemand(Commodity,HouseholdH) 
      =G= 
     ComPrice(Commodity) 
      *AdjConsumerQ(Commodity) 
        *AddAdjCP(Commodity) 

   + ComPrice(Commodity) 
        *AddAdjCP(Commodity) 
           *(Gamma(Commodity,HouseholdH)) 
   + ( Beta(Commodity,HouseholdH) 
         *(  HhExpend(HouseholdH) 
           - SUM(CommodityP,ComPrice(Commodity) 
                  *AddAdjCP(Commodity) 
                    *Gamma(CommodityP,HouseholdH) ) 
          ) )  ; 

 
Government Sector 
 

Six equations contribute to the government sector are the government demands for goods 
and services, consumption, income, expenditure, transfer payments, and saving. 

 
Government Demand 
  

The government quantity or value of production in government sector followed equation 
(32) on page 28 in SGM model documentation is shown below: 
 
GovDemandEq(HouseHoldG).. 
    GovDemand(HouseHoldG) 
      =E= 
    (   GovIncome 
      - GovSaving 
      - GovTrnsfPaymt 
    )*GovShare0(HouseholdG) 
       *(Pgss0(HouseholdG) 
          **(GammaGov-1)) 
              *( sum(HouseholdGP, 
                ( GovShare0(HouseholdGP) 
                     *Pgss0(HouseholdGP) 
                        **(GammaGov) )) )**(-1)   ; 
 
Government Consumption by Sector 
 
 For the government consumption by sector, the fixed weights for the government 
expenditure and the share of tax revenue among sectors and households are used. 
 
GovConsumpEq(Sector,HouseHoldG).. 
   GovConsumption(Sector,HouseHoldG) 
     =E= 
   GovFixWeight(Sector,HouseHoldG)*GovShareTR0("GoToSector") 
    * (GovIncome - GovSaving) 
        /Comprice(Sector)   ;. 

Government Income 
 

The government income is a sum of corporate tax, personal income tax, social security 
tax, indirect business tax, and carbon tax.  Because of space limitations the expression of this 
equation is shown in lines 8308 to 8415. 

 
Government Budget 
 

The government budget is used to ensure that the government income and expenses are 
exhausted (balance).  This constraint is expressed as: 

 
GovBudgetEq.. 
    GovIncome 
     =E= 
    GovSaving 
  + GovTrnsfPaymt 
  + sum((Sector,HouseHoldg), 
            GovConsumption(Sector,HouseHoldG) 
              *Comprice(Sector) 
       )     ; 

 
Government Transfer Payments 
 

The government transfer payments can be modeled in two ways.  As shown in equation 
(28) in SGM documentation, the transfer payments are assumed to be a function of population 

1
0

ββ PopulationTransfer =  
where β0 and β1 are given.  Because the current SGMGAMS is a static model (populations 
remains constant), with this function the government transfer payments will be exogenous 
regardless of changes in the government consumption.  Therefore, the second alternative model 
is used.  This alternative the government transfer payments is modeled as a function of 
expenditure and consumption: 
 
GovTrnsfEq.. 
    GovTrnsfPaymt 
      =E= 
    GovExpend 
  - sum((Sector,HouseHoldg), 
             GovConsumption(Sector,HouseHoldG)*Comprice(Sector)        ) 
      ; 
Government Expenditure 
 

The government expenditure is expressed as: 
 

GovExpendEq.. 
    GovExpend 
      =E= 

    GovTrnsfPaymt 
  + sum((Sector,HouseHoldg), 
              GovConsumption(Sector,HouseHoldG) 
                *Comprice(Sector) 
       ) 
        ; 

 
Investment Sector 
 

The investment demand function for sectors follows equation (21) on page 24 in the 
SGM model documentation while the investment demand function for subsector using the logit 
function follows equation (22) on page 25 in the SGM model documentation.  Note that crude oil 
and natural gas sectors have exogenous investment. 

The model will solve for investment based on the expected profit rate.  The expected 
profit rate is a function of prices and quantities as described in equations (7) to (15) in the SGM 
model documentation: 

A set of equations involves with the investment sectors are shown from lines 7858 to 
8206. 

 
Expected Profit Rate 
 

Since the calculation of the expected profit rate is complicated, only main equations are 
presented below.  The rest of the equations are shown from lines 7858 to 8206.  Information to 
calculate the present values used in the model including investment lifetime, real interest rate, 
sector interest rate, and wedge rate are directly obtained from SGM.   

* #### Sector expected profit rate 
 
ExpProfSecEq(AllSecAllSub) 
  $(     YesInv(AllSecAllSub) 
     and OnlySector(AllSecAllSub) 
     and (1 - InvTaxCredit(AllSecAllSub) ) 
   ).. 
 
    ExpProfSec(AllSecAllSub) 
      =E= 
   ( TechAlpha0(AllSecAllSub) 
       * PriceReceive(AllSecAllSub) 
           * PresentValue(AllSecAllSub) 
              *(  Zvalue(AllSecAllSub)**(1/RRho(AllSecAllSub)) 
                ) 
                * ( (AlphaInv("Capital",AllSecAllSub) 
                     ) 
                        **(1/Rho(AllSecAllSub)) ) 
    )/(1 - InvTaxCredit(AllSecAllSub) ) 
     ; 
 
* #### SubSector expected profit rate 
 
ExpProfSubEq(AllSecAllSub) 
  $( YesInv(AllSecAllSub) 
     and OnlySubSector(AllSecAllSub) 
     and (not sameas(AllSecAllSub,"SElecBiom")) 
     and (1 - InvTaxCredit(AllSecAllSub) ) 
   ).. 
 
    ExpProfSub(AllSecAllSub) 
     =E= 
   ( 
 
     (( ( 1/sum(SubSectorP 
               $(     (not sameas(SubSectorP,"SElecNucl")) 
                  and (not sameas(SubSectorP,"SElecHydro")) 
                  and (not sameas(SubSectorP,"SElecBiom")) 
                 ), 
            SubInvSh3(SubSectorP) 
           ) 
        ) 
      )* 
           (  SubInvSh1(AllSecAllSub) 
                 /SubInvSh1("SElecOil") )**LamdaInv                        ) 

        $(not sameas(AllSecAllSub,"SElecOil")) 
    + ( 1/sum(SubSectorP 
               $(     (not sameas(SubSectorP,"SElecNucl")) 
                  and (not sameas(SubSectorP,"SElecHydro")) 
                  and (not sameas(SubSectorP,"SElecBiom")) 
                 ), 
            SubInvSh3(SubSectorP) 
           ) 
       )$sameas(AllSecAllSub,"SElecOil") 
 
   )/(1 - InvTaxCredit(AllSecAllSub) ) 
     ; 

 
Investment Demand 
 

The investment takes place at the sector and subsector.  Also, some sectors (crude oil and 
natural gas) have fixed investment.  The investment is defined as a function of last period’s 
investment, a base rate, the growth in working population, expected profit rate, and elasticity of 
excess profit rate.  These parameters are given in SGM, except for the expected profit rate that is 
determined within the model as described above. 
 
InvestDem(AllSecAllSub) 
 $ (  (    YesInv(AllSecAllSub) 
        or OnlySector(AllSecAllSub) 
        or OnlySubSector(AllSecAllSub) 
       ) 
    and (not sameas(AllSecAllSub,"SElecBiom")) 
   ).. 
 
    QInvest(AllSecAllSub) 
      =E= 
 
* #### this is for Sector 
 
    ( QInvest0("1985",AllSecAllSub) 
      * BaseRate 
         * Population0("1990","WorkRate") 
           * ( ExpProfSec(AllSecAllSub))**ElasInv(AllSecAllSub) 
    )$( OnlySector(AllSecAllSub) and not ExoInv(AllSecAllSub) ) 
 
 
* #### this is for SubSector using logit sharing structures 
 
  + ( QInvest0("1985",AllSecAllSub) 
      * BaseRate 
         * Population0("1990","WorkRate") 

           * (ExpProfSub(AllSecAllSub))**ElasInv(AllSecAllSub) 
    )$(       OnlySubSector(AllSecAllSub) 
          and (not ExoInv(AllSecAllSub)) 
          and YesInv(AllSecAllSub) 
       ) 
 
* #### this is for exogenous investment 
 
  + (QExogInv(AllSecAllSub))$( ExoInv(AllSecAllSub)  ) 
      ; 
 
Capital Stock 
 

The investment will be converted into a capital stock for the next time period.  Capital 
stocks are modeled as falling into 4 vintage classes.  For the next time period, the new vintage of 
capital stocks operates under the long-run elasticity of substitution while the old vintage of 
capital stocks operates under the short-run elasticity of substitution.   

 
KStockEq(AllSecAllSub,Vintage2) 
  $(not sameas(AllSecAllSub,"SElecBiom")).. 
 
    KStock(AllSecAllSub,Vintage2) 
      =E= 
   (   2*QInvest0("1985",AllSecAllSub) 
     + 3*QInvest(AllSecAllSub) 
   )$(sameas(Vintage2,"0")) 
 
   + KStock0(AllSecAllSub,"-1") 
        $(sameas(Vintage2,"-1")) 
   + KStock0(AllSecAllSub,"-2") 
        $(sameas(Vintage2,"-2")) 
   + KStock0(AllSecAllSub,"-3") 
        $(sameas(Vintage2,"-3")) 
    ; 

 
Exports and Imports 
 

Because the current model is a single country operation (only the US), the quantity of 
exports and imports are defined as a fixed proportion to the production.  The proportion values 
are calculated from quantity exports and imports given in SAM table. 

 
ImportEq(Sector,SubSector) 
  $( YesImport(Sector) 
     and TisMap(Sector,SubSector) 
   ).. 

    QImport(Sector,SubSector) 
      $TisMap(Sector,SubSector) 
      =E= 
   (  QImportPct(Sector) 
      *sum(Vintage,Production(Sector,SubSector,Vintage)) 
    )$(not sameas(Sector,"ElecGen")) 
 
 +  (  QImportPct(SubSector) 
      *sum(Vintage,Production(Sector,SubSector,Vintage)) 
    )$(sameas(Sector,"ElecGen")) 
       ; 
 
ExportEq(Sector,SubSector) 
  $( YesExport(Sector) 
     and TisMap(Sector,SubSector) 
   ).. 
 
    QEXport(Sector,SubSector) 
      $TisMap(Sector,SubSector) 
      =E= 
   (  QExportPct(Sector) 
      *sum(Vintage,Production(Sector,SubSector,Vintage)) 
    )$(not sameas(Sector,"ElecGen")) 
 
 +  (  QExportPct(SubSector) 
      *sum(Vintage,Production(Sector,SubSector,Vintage)) 
    )$(sameas(Sector,"ElecGen")) 
       ; 
 
Note that when the current model moves from the single region model to global model, these 
exports and imports will be modified to be traded at world market prices. 

 
Greenhouse Gas Modeling 

 
Modeling greenhouse gasses involves portrayal of both emissions and sinks from all 

sectors.  We discuss separately the agricultural sector and from energy sectors as well as non-
energy sectors.  The greenhouse gas limitation is used to cap on the carbon net emission under 
the Kyoto. 

 
Agricultural Emissions and Sinks 

 
Agricultural emissions and sinks are included in the model using response functions 

estimated using data generated from repeated runs of the U.S. Agricultural Sector Model 
greenhouse gas version (ASMGHG).  These are used to model agricultural emissions and sink 
reactions to changes in the general economy.  Conceptually, these functions are modeled as 
follows: 

Agricultural Emissions = exp( a1  
                                      + a2 * ln(e1+Carbon price)  

+  a3 * ln(e2*(1+ in US agricultural demand)) ∆
+  a4 * ln(e3*(1+ in Energy price))  ∆
+  a5 * ln(e4 *(1+ in US agricultural export demand))) ∆
 

Ln(Agricultural Sinks)   = exp( b1  
                                      + b2 * ln(e1+Carbon price)  

+  b3 * ln(e2*(1+ in US agricultural demand)) ∆
+  b4 * ln(e3*(1+ in Energy price))  ∆
+  b5 * ln(e4 *(1+ in US agricultural export demand))) ∆
 

where ai and bi are estimated parameters associated with carbon price, US agricultural demand, 
Energy price, and US exports; and ei are the base values (0 for carbon price and 100 for the 
others).  These estimated parameters (ai and bi) are previously estimated in the response function 
with a log-linear function, ln(Y) = A + β*ln(x) where A and β are a vector of intercept terms and 
a vector of estimated parameters associated with a vector of Y and x, respectively.  The base 
functions with all of the independent variables held at the base level depict the ASMGHG output 
under a zero carbon price, and the1997 levels of energy price, domestic agricultural product 
demand, and agricultural export demand.  The completed expression of these equations is 
presented in lines 8503 to 8597. 
 
Energy/non-energy Sector Emissions 

 
To calculate emissions by the energy sectors, the production levels within the sectors 

using energy inputs are converted into physical energy units (joules) using the conversion from 
SGM and then these physical energy units are multiplied by GHG emission coefficients as 
shown below: 

 
Emissions =    ci * gig *values of production 

where ci is the physical energy conversion and gig is the emission coefficient by Ghg and energy 
input types (see lines 8598 to 8632). 
 
   Conversion (ci)  GHG emission coefficients (gig) 

       CO2  CH4  N2O 
 
Crude oil  0.000436   18.81  0.187  0.010 
Natural gas  0.000436   14.28  0.187  0.007 
Coal   0.001045   23.74  0.354  0.014 
 
Net Emissions 
 

This equation implies that the net emissions in the region are equal to the sum of the 
agricultural and energy emissions minus the agricultural sinks. 

NetEmisQEq(Region,Ghg)$sameas(Region,"US").. 
* #### Sources and Sinks from Ag sector 
 
     + sum(Vintage, 
                AgEmitQ(Ghg,Vintage) 
             -  AgSinkQ(Ghg,Vintage)$(not sameas(Ghg,"N2O")) 
          ) 
 
* #### Sources from Energy sector 
 
     + sum((Sector,Vintage) 
        $TisEnergy(Sector), 
            EnergyEmitq(Sector,Ghg,Vintage) 
               $( CEmission(Sector,Ghg)              ) 
          ) 
        =G= 
     NetEmiTQ(Region,Ghg) 
        ; 
 
GHG Emissions Limitation 
 

At the base, the amount of greenhouse gas is not limited and therefore the US net 
emissions are in the neighborhood of the 1990 net emissions and the carbon price is zero.  When 
the amount of greenhouse gas is limited at some levels (i.e. Kyoto), the model will search for a 
carbon price at the equilibrium to ensure that this greenhouse gas constraint is binding.  We have 
done trial runs at the Kyoto limits and get a carbon price from $12 to $50 per a metric ton of 
carbon equivalent (see output.txt file). 

 
EmisQLimitEq(Region)$sameas(Region,"US").. 
  sum(TisLevel, 
        sum(Ghg,GhgLimitQ(Region,Ghg,TisLevel) )    ) 
  + 10*(1-LimitType) 
   =G= 
  sum(Ghg,NetEmiTQ(Region,Ghg)) 
   ; 

 
Carbon Tax Rate 
 

Imposing a carbon net emission cap imposes a cost on sectors that use inputs involving 
with the greenhouse gas.  Therefore, the carbon tax rate is first calculated and then it is used to 
calculate the total carbon tax value which is considered as revenues to the government but costs 
to sectors.  This total carbon tax values is then included into zero profit condition as costs.   

 
CarbTaxRateEq(Sector,Ghg) 
  $TisEnergy(Sector).. 

  CarbTaxRate(Sector,Ghg) 
   =E= 
  Pcarb("US") 
   * PricToPhysic(Sector) 
       * CEmission(Sector,Ghg) 
           *GWP(Ghg) 
             *Exchange0 
  ; 
 
* #### Carbon tax revenues by users 
 
CarbTaxTotEq(Sector,Ghg) 
  $TisEnergy(Sector).. 
 
   CarbTaxTot(Sector,Ghg) 
     =E= 
   CarbTaxRate(Sector,Ghg) 
          * ( 
                 + sum(MapSecSub(OthSector,SubSector), 
                         QIntC(Sector,OthSector,SubSector) 
                            $ica1(Sector,OthSector,SubSector)) 
                              *ComPrice(Sector) 
 
                 + sum(HouseholdH, 
                         GoodsDemand(Sector,HouseholdH) 
                            $SAM(Sector,HouseholdH)  ) 
                               *ComPrice(Sector) 
 
                 + sum(HouseHoldG, 
                         GovConsumption(Sector,HouseHoldG) 
                           $GovConsumption0(Sector,HouseHoldG) 
                      )*ComPrice(Sector) 
 
                 + sum(MapSecAllSub(Sector,AllSecAllSub), 
                         QInvest(AllSecAllSub)$YesInv(AllSecAllSub)) 
                            *ComPrice(Sector) 
                 + QStockChnge(Sector)*ComPrice(Sector) 
 
                 + sum(MapSecSub(Sector,SubSector)$YesExport(Sector), 
                         QExport(Sector,SubSector) 
                            *PExp(Sector)) 
             ) 
    ; 

Equilibrium Conditions 
 

A set of non-zero prices in output and input markets, production, consumption levels, 
production levels, and factor usages constitutes an economic equilibrium solution and a solution 
to a CGE of the situation if the following conditions are satisfied. 
 
Factor Market Balance 

 
Total factors usage in production is less than or equal to the total supply in every factor input 

markets (labor and capital) which is composed from the household endowments. 
 
FactorMkt(Factor2,Sector,SubSector) 
  $TisMap(Sector,SubSector).. 
 
  sum(HouseholdH, 
    Endowment1(Factor2,Sector,SubSector,HouseholdH) ) 
      $sameas(Factor2,"Labor") 
 + sum(Vintage2,KAPrior0(Sector,SubSector,Vintage2)) 
    $sameas(Factor2,"Capital") 
     =G= 
   FactorQ(Factor2,Sector,SubSector) 
   ; 
 
Commodity Market Balance 
 
 This condition implies that the total demand in every output market including consumer 
and intermediate production usage is less than or equal to total supply in that market.  In other 
words, the excess demand in each output market is less than or equal to zero: 
 
SupplyDemandEq(Sector) 
  $(      COMPRICE0(Sector)    ).. 
 
   sum(MapSecSub(Sector,SubSector), 
     sum(Vintage,Production(Sector,SubSector,Vintage) )) 
   =G= 
  sum(HouseholdH, 
        GoodsDemand(Sector,HouseholdH) 
           $SAM(Sector,HouseholdH)  ) 
 
  + sum(HouseHoldG, 
         GovConsumption(Sector,HouseHoldG) 
        ) 
 
  +  sum(MapSecSub(OthSector,SubSector) 
           $ica1(Sector,OthSector,SubSector), 
            QIntC(Sector,OthSector,SubSector) ) 

  + sum(MapSecAllSub(Sector,AllSecAllSub), 
          QInvest(AllSecAllSub)$YesInv(AllSecAllSub)         ) 
 
  + sum(MapSecSub(Sector,SubSector) 
          $YesExport(Sector), 
            QExport(Sector,SubSector)) 
 
  - sum(MapSecSub(Sector,SubSector) 
          $YesImport(Sector), 
            QImport(Sector,SubSector)) 
 
  + QStockChnge(Sector) 
 
       ; 
 
Zero Profit Condition 

 
For each production sector revenues are less than or equal to costs with in effect all rents 

allocated to factors.  Thus the total revenue is less than factor usage costs plus costs of intermediate 
products obtained plus other tax payments and plus carbon costs when the carbon emission limits is 
imposed. 
 
Profit(Sector,SubSector,Vintage) 
  $TisMap(Sector,SubSector).. 
 
* #### Factor costs 
      sum(Factor2, 
            FactorQ(Factor2,Sector,SubSector) 
               * FactPriceSec(Factor2,Sector,SubSector) 
         ) 
 
* #### Intermediate inputs costs 
   + ComPrice(Sector) 
         * sum(Activity 
               $ica1(Activity,Sector,SubSector), 
                   QIntC(Activity,Sector,SubSector) ) 
 
* #### corporate taxes 
   + sum(MapSecSubSub(Sector,AllSecAllSub,SubSector), 
            1 - PriceReceive(AllSecAllSub) 
         ) 
           * Production(Sector,SubSector,Vintage) 
 
* #### Carbon costs from Agricultural sector from ASM 
   + sum(Region$sameas(Region,"US"), 

          sum(Ghg, 
                   PCarb(Region)*PricToPhysic("OthAg")* 
                       (   AgEmitQ(Ghg,Vintage) 
                         - AgSinkQ(Ghg,Vintage)$(not sameas(Ghg,"N2O")) 
                        ) 
              ) 
         )$AgGroup(Sector) 
 
* #### Carbon costs from Energy sector 
         +  sum(ghg, 
                  sum(MapSecSub(OthSector,SubSector)$TisEnergy(OthSector), 
                         CarbTaxRate(OthSector,Ghg)*QIntC(OthSector,Sector,SubSector) 
                            $ica1(OthSector,Sector,SubSector) 
                              *ComPrice(OthSector)) 
                 ) 
         +  sum(Ghg, 
                 CarbTaxRate(Sector,Ghg)$TisEnergy(Sector) 
                    *  (   QInvest(Sector) 
                         + QStockChnge(Sector) 
                        )$YesInv(Sector) 
                            *ComPrice(Sector) 
                ) 
         +  sum(Ghg, 
                 sum(MapSecSub(Sector,SubSector)$YesExport(Sector), 
                      CarbTaxRate(Sector,Ghg)$TisEnergy(Sector) 
                         * QExport(Sector,SubSector) 
                            *PExp(Sector)  ) 
               ) 
   =G= 
  ComPrice(Sector) 
     * Production(Sector,SubSector,Vintage) 
          ; 
 
Price Block 
 

Below equations represent the price system of the model.  Some of these price equations 
are linked to other prices variables while some are linked to non-price variables in the model. 

 
Intermediate Input Prices 

 
The intermediate input prices show costs of disaggregated intermediate inputs. 

 
PIntAEq(Sector,SubSector) 
  $TisMap(Sector,SubSector).. 
 
  PIntA(Sector,SubSector) 

    =G= 
  SUM(Activity, 
       ica1(Activity,Sector,SubSector) 
           *ComPrice(Activity)        ) 
    ; 
 
Value-Added Prices 

 
This value-added price indicates that the total revenue (net of taxes) is fully exhausted by 

payments for value-added and intermediate inputs.   
 
PVAEq(Sector,SubSector) 
   $TisMap(Sector,SubSector).. 
 
    sum(MapSecSub(Sector,SubSector), 
     ComPrice(Sector) 
     ) 
          *  sum(Vintage,Production(Sector,SubSector,Vintage)  ) 
     =G= 
    PValAdd(Sector,SubSector)* QValAdd(Sector,SubSector) 
  + PIntA(Sector,SubSector)*  QIntA(Sector,SubSector) 
     ; 
 
Price Paid 

 
The price paid represents the price in which sectors paid for the use of input i. 

 
PricePaidEq(Activity,AllSecAllSub) 
  $( (not sameas(AllSecAllSub,"SElecBioM")) 
     and (not sameas(Activity,"IBT")) 
     and (not sameas(Activity,"Land")) 
   ).. 
 
   PricePaid(Activity,AllSecAllSub) 
     =E= 
   ComPrice(AllSecAllSub) 
       $(     (not sameas(Activity,"Labor") ) 
          and (not sameas(Activity,"Capital") ) 
          and (not sameas(Activity,"Land") ) 
           and OnlySector(AllSecAllSub) 
         ) 
 + ComPrice("ElecGen") 
       $(     (not sameas(Activity,"Labor") ) 
          and (not sameas(Activity,"Capital") ) 
          and (not sameas(Activity,"Land") ) 
           and OnlySubSector(AllSecAllSub) 

         ) 
  + sum(MapSecSub(AllSecAllSub,SubSector)$TisMap(AllSecAllSub,SubSector), 
             FactPriceSec(Activity,AllSecAllSub,SubSector) 
         $(  (  sameas(Activity,"Labor") 
                or sameas(Activity,"Capital") 
              ) 
           and OnlySector(AllSecAllSub) 
           and YesCESVa(AllSecAllSub) 
     )     ) 
  + FactPriceSec(Activity,"ElecGen",AllSecAllSub) 
         $(  (  sameas(Activity,"Labor") 
                or sameas(Activity,"Capital") 
              ) 
           and OnlySubSector(AllSecAllSub) 
           and YesCESVa(AllSecAllSub)     ) 
  + ( sum(Ghg$(TisEnergy(Activity)    ), 
              CarbTaxRate(Activity,Ghg))       ) 
         $  (  sameas(Activity,"Labor") 
                or sameas(Activity,"Capital") 
                or TisEnergy(Activity)              ) 
  + ( sum(Ghg$(TisEnergy(Activity) 
                and OnlySubSector(AllSecAllSub)              ), 
              CarbTaxRate("ElecGen",Ghg))       ) 
         $  (  sameas(Activity,"Labor") 
                or sameas(Activity,"Capital")              ) 
       ; 
 
Price Received 

 
The price received represents the price in which producers received from the sales of 

output j excluding indirect business taxes. 
 
PriceReceivEq(AllSecAllSub) 
  $(not sameas(AllSecAllSub,"SElecBiom")).. 
 
   PriceReceive(AllSecAllSub) 
     =E= 
    sum(MapSecAllSub(Activity,AllSecAllSub), 
         PricePaid(Activity,AllSecAllSub) 
          *(1 - IBTTaxRate0(AllSecAllSub)) 
      ) 
     ; 
 
Factor Price by Sector 

This equation represents a factor price by sector and subsector.  Note that the average 
nation wage factor price (FactPriceNation) is an endogenous variable where the activity specific 
“wage-distortion factor (FactPriceDist1) is exogenous. 
 
FactPriceSecEq(Factor,Sector,SubSector) 
    $(     TisMap(Sector,SubSector) 
       and YesCESVa(Sector) 
      ).. 
 
   FactPriceSec(Factor,Sector,SubSector) 
   =E= 
   FactPriceNation(Factor) 
     * FactPriceDist1(Factor,Sector,SubSector) 
   ; 
 
Technical Change 
 

We assume no technical change for the current model but can add this when we add a 
dynamic updating procedure in a multi-time period version. 
 
Model Solve Statements 

 
The class of models that this SGMGAMS falls into is the so-called mixed 

complementarity class (an MCP) in GAMS.  To solve such models we must have a model with 
complementarity requirements associated with each equation as shown in the table below.  The 
complementary relationships are expressed in the model statement. 
 
Complementary Items 
 

Variable Name Equation Name 
  
Quantity block:  
QValAdd(Sector,SubSector) CESQVAEq(Sector,SubSector) 
FactPriceNation(Factor) CESQVAFOC(Factor) 
QintA(Sector,SubSector) QintAEq(Sector,SubSector) 
QIntC(Activity,Sector,SubSector) QIntCEq(Activity,Sector,SubSector) 
PImp(Commodity) ImpBal(Commodity) 
PExp(Commodity) ExpBal(Commodity) 
QImport(Sector,SubSector) ImportEq(Sector,SubSector) 
QExport(Sector,SubSector) ExportEq(Sector,SubSector)   
Price block:  
FactPriceSec(Factor,Sector,SubSector) FactPriceSecEq(Factor,Sector,SubSector) 
PValAdd(Sector,SubSector) PVAEq(Sector,SubSector) 
PIntA(Sector,SubSector) PIntAEq(Sector,SubSector) 
PricePaid(Activity,AllSecAllSub) PricePaidEq(Activity,AllSecAllSub) 
PriceReceive(AllSecAllSub) PriceReceivEq(AllSecAllSub) 

Household consumption:  
HhExpend(HouseholdH) HhExpendEq(HouseholdH) 
Hhincome(HouseholdH) IncomeEq(HouseholdH) 
GoodsDemand(Commodity,HouseholdH) GoodsDemandEq(Commodity,HouseholdH)   
Investment:  
QInvest(AllSecAllSub) InvestDem(AllSecAllSub) 
AlphaInv1(Activity,AllSecAllSub) AlphaInv1Eq(Activity,AllSecAllSub) 
AlphaInv2(AllSecAllSub)        AlphaInv2Eq(AllSecAllSub)        
AlphaInv3(Activity,AllSecAllSub) AlphaInv3Eq(Activity,AllSecAllSub) 
AlphaInv(Activity,AllSecAllSub)  AlphaInvEq(Activity,AllSecAllSub)  
IBigAij(Activity,AllSecAllSub)   IBigAijEq(Activity,AllSecAllSub)   
Zvalue(AllSecAllSub) ZvalueEq(AllSecAllSub) 
SubInvSh1(AllSecAllSub) SubInvSh1Eq(AllSecAllSub) 
SubInvSh3(AllSecAllSub) SubInvSh3Eq(AllSecAllSub) 
ExpProfSec(AllSecAllSub) ExpProfSecEq(AllSecAllSub) 
ExpProfSub(AllSecAllSub) ExpProfSubEq(AllSecAllSub) 
Kstock(AllSecAllSub,Vintage) KstockEq(AllSecAllSub,Vintage)   
Government:  
GovSaving GovBudgetEq 
GovIncome GovIncomeEq 
GovTrnsfPaymt GovTrnsfEq 
GovDemand(HouseHoldG) GovDemandEq(HouseHoldG) 
GovConsumption(Sector,HouseHoldG) GovConsumpEq(Sector,HouseHoldG) 
GovExpend GovExpendEq   
Greenhouse gas:  
AgEmitQ(Ghg,Vintage) AgEmitQEq(Ghg,Vintage) 
AgSinkQ(Ghg,Vintage) AgSinkQEq(Ghg,Vintage) 
EnergyEmitQ(Sector,Ghg,Vintage) EnergyEmitQEq(Sector,Ghg,Vintage) 
NetEmiTQ(Region,Ghg) NetEmisQEq(Region,Ghg) 
PCarb(Region) EmisQLimitEq(Region) 
CarbTaxRate(Sector,Ghg) CarbTaxRateEq(Sector,Ghg) 
CarbTaxTot(Sector,Ghg) CarbTaxTotEq(Sector,Ghg)   
Equilibrium Condition:  
FactorQ(Factor,Sector,SubSector) FactorMkt(Factor,Sector,SubSector) 
ComPrice(Sector) SupplyDemandEq(Sector) 
Production(Sector,SubSector,Vintage) Profit(Sector,SubSector,Vintage) 

 
Model Files 
 
To run the model, please open the integrating file called allsgm.gms.  Allsgm.gms includes  
 

sgmdat.gms  containing the SGMGAMS sets, parameters, scalars, and data 

sgmcalib.gms calculating some predetermined parameters needed to calibrate the 
model and other exogenous values 

sgmmod.gms  containing the SGMGAMS equation structure 
sgmparm.gms containing declaration of sets and parameters used for report 

writing 
sgmreport.gms containing report writing code 
sgmcompr.gms containing report writing code for comparative analysis 
sgmloop.gms containing codes for comparative analysis and writing out 

comparative results.  Here, emission reductions from the 
agriculture are considered. 

sgmloopfix.gms containing codes for comparative analysis and writing out 
comparative results.  Here, emission reductions from the 
agriculture are not considered. 

 
Algorithm Used 
 

The SGMGAMS is best solved with the PATH solver.  Details on that solver can be 
found at http://www.gams.com/solvers/solvers.htm#PATH.  So we choose PATH as the solver 
and solve with  
 

OPTION MCP = PATH; 

SOLVE SGMCGE USING MCP; 
 
Resultant Solution 
 

In turn, a solution arises and it can be used in the normal report writing, graphics etc. as with 
any other model (see sgmgams.out file) 

